

EXCEL

9

Text Functions, Conditional Functions, and Formula Auditing



In this chapter, you will learn various functions that give you greater ability for analysis and decision making. You will create functions that sum or count values that meet your desired criteria. You will also explore functions used to clean up and rearrange text on your worksheet, as well as learn what you can do to find and prevent formula errors in your calculations.

LEARNING OBJECTIVES

- ▶ Use functions to format text
- ▶ Create conditional functions using IF and IFS criteria
- ▶ Create formulas using nested functions
- ▶ Find and correct errors in formulas
- ▶ Use 3-D cell references in formulas

Project: Analyzing Sales Information

The Airspace Travel monthly sales results are in, and the data has been compiled for all company agents and managers in a worksheet for your review. Because the data was imported from different sources, you need to clean up the text entries. You'll also use various conditional functions to pull out important information about specific performance.

Using Functions to Modify Text

Workbook data that comes from sources other than Excel may be formatted incorrectly. Data may also have been entered by multiple users, each using a different method of data entry. For example, one person might enter names into a worksheet using all capital letters, and another person might capitalize the first letter of the name only. Then when the two worksheets are combined, name entries will not be consistent. Another problem can occur when data is either entered in too few or too many columns, such as entering the first and last names together in one column when it is better to enter this data in separate columns.

Tip!

When storing data, it's best practice to use the smallest individual units for each field of information. In Excel, this means each column.

Although many people primarily think of Excel as a way to work with numbers, there are quite a few functions that allow you to work with text as well. You can use text functions to fix the issues mentioned or to manipulate text data to be used for a different purpose. There are functions that let you change case, combine or separate text, remove spaces, and extract or even replace text.

Changing Case

PROPER, UPPER, and LOWER are three functions that allow you to change the case of the input text. PROPER converts the first letter of each word to uppercase (capital) and all other letters to lowercase. As you can probably guess, UPPER converts all letters to uppercase and LOWER converts all letters to lowercase.

When using these functions, the function argument is simply the text to convert; you can use a cell reference or the text itself. For example, cell A2 shows text where some words are in all lowercase letters and another word is in all uppercase letters. The formula displayed in cell B2, =PROPER(A2), tells Excel to capitalize the first letter of each word from cell A2, and the result is *Use Your Imagination* (displayed in cell C2).

	A	B	C
1	Text	Formula	Result
2	use your IMAGINATION	=PROPER(A2)	Use Your Imagination
3		=LOWER("AND")	and
4	make some magic!	=UPPER(A4)	MAKE SOME MAGIC!

Extracting Text

In some cases, only a part of the cell's contents is needed, or there may be extra characters or spaces you don't want. The LEFT, MID, and RIGHT functions extract a certain number of characters from the **text string**. The TRIM function removes all spaces except for a single space between words.

The LEFT and RIGHT functions take two arguments: the text (which can be actual text or a cell reference) and the number of characters to extract.

	A	B	C
1	Text	Formula	Result
2	BASKabcdefg	=LEFT(A2,4)	BASK
3	abcdefgETB	=RIGHT(A3,3)	ETB
4	abcALLdefg	=MID(A4,4,3)	ALL
5	Who likes basketball?	=TRIM(A5)	Who likes basketball?

The TRIM function's only argument is the text from which to remove the spaces.

The MID function requires three arguments: the text, the position of the first character to extract, and then the number of characters to extract.

Merge and Modify Text with Functions and Flash Fill

There are several ways to merge, or concatenate, text in Excel. The CONCAT function, which replaces the CONCATENATE function from earlier versions of Excel (though it's still available for compatibility), allows you to combine two or more separate text entries or a range of text entries into one cell. TEXTJOIN is another function that combines text; it also inserts a character between each entry, called a delimiter.

Flash Fill can also be used to combine text, and it has other advantages too. Flash Fill can combine multiple entries into one cell or extract text from one text string into multiple entries, and it can perform many other tasks. After you have entered one or two examples, Flash Fill looks for patterns in your data entries and automatically fills in the remaining values.

Tip!

The CONCAT and TEXTJOIN functions are used as part of a formula, whereas Flash Fill is just a tool that enters text values into cells.

For example, if you have a column with First Name and another column with Last Name, CONCAT or TEXTJOIN can be used to combine the two names into one cell. Flash Fill can do this but can also do the opposite task; take one name and separate it into First and Last columns. You could use Flash Fill to extract one part of the cell only, such as the first three letters of the last name, or to extract the area code from a phone number. Flash Fill can even append or insert text, such as automatically creating email addresses from a list of employee names.

One big difference is that Flash Fill uses adjacent data only, whereas CONCAT and TEXTJOIN use a cell or range reference so the text could be anywhere on the worksheet or even on another

worksheet. Another difference is that a function will update automatically if changes are made to the source of the text, but after Flash Fill is used the text becomes static.

Column C uses CONCAT to combine the text from columns A–B into one cell.


	A	B	C	D
1	First Name	Last Name	Name	
2	Jaime	Burgess	Jaime Burgess	
3	Ashley	Bradford	Ashley Bradford	
4	Deborah	Secrett	Deborah Secrett	
5	Steven	Samuel	Steven Samuel	

The function arguments are cell references or text. The second argument (" ") inserts the text within the quotation marks, in this case a space, between the first and last names from column A and B.

The original data is in column A.

	A	B
1	Phone Number	Area Code
2	232-555-1023	232
3	416-555-1024	416
4	905-555-1025	905
5	771-555-1026	771

The area code 232 was manually entered in cell B2 while Flash Fill was used to fill in the others.



The Flash Fill Options button includes options to Accept and Undo the suggested entries.

☰ Data→Data Tools→Flash Fill | Home→Editing→Fill→Flash Fill

Other Text Functions

Be sure to take some time to explore the other text functions available in Excel. There are text functions that allow you to replace or substitute text within a text string, functions for finding the text's position, and functions to calculate text length. You can even insert a function that will repeat a text character a specified number of times. These are just some examples; there is a long list of functions in the Text category of Excel's function library.

TEXT FUNCTIONS

Function	Description	Example
REPLACE	Replaces part of a text string with another text string, such as replacing digits in a credit card number to display 8181-xxxx-xxxx-1188	Cell B1: 8181-3011-1103-1188 Formula: =REPLACE(B1,6,9,"xxxx-xxxx") Result: 8181-xxxx-xxxx-1188
SUBSTITUTE	Looks for an exact match (case-sensitive) and replaces old text with new text if found, such as replacing Mgr with Manager	Cell B4: Mgr Formula: =SUBSTITUTE(B4,"Mgr","Manager") Result: Manager

TEXT FUNCTIONS

Function	Description	Example
LEN	Determines the number of characters in a cell entry	Cell B7: 2223334444 Formula: =LEN(B7) Result: 10
REPT	Repeats text, such as the letter A five times	Formula: =REPT("A",5) Result: AAAAA

☰ Formulas→Function Library→Text

DEVELOP YOUR SKILLS: E9-D1

In this exercise, you will use text functions to clean up the text entries in the *Airspace Sales Results* worksheet.

1. Start Excel, open **E9-D1-Sales** from your **Excel Chapter 9** folder, and save it as: **E9-D1-SalesAnalysis**

Change Case

The workbook opens to the *Jul Sales* sheet. Notice that the names in column A are not consistent in terms of the capitalization used. You want to correct that and, to begin, you will create a blank column in which the names can be converted to proper capitalization.

2. Insert a new column to the left of **column A**.
3. In **cell A5**, enter the formula **=PROPER(B5)** and then fill the formula down the column to **cell A33** and **AutoFit** the column width.

The cells now display the text with the first letter of each name capitalized. To keep just the text and not the function, you will copy and paste the values only to the same range.

4. With the **range A5:A33** selected, choose **Home→Clipboard→Copy** .
5. Without changing the selection, choose **Home→Clipboard→Paste menu button** ▼ → **Values**  to paste the values only (not the formulas) into the selected range.

Instead of the *PROPER* function in the Formula Bar for cell A5, you should now see the text Amir Johnson, with both names capitalized.

6. Delete the names from the **range B5:B33**.

Extract Text

7. Insert a column to the left of **column E**.
You will extract the first six characters of the employee ID numbers, removing the extra characters.
8. In **cell E5**, enter the function **=LEFT(D5,6)** and then fill the formula down the column.
9. With the **range E5:E33** still selected, copy the formulas and then paste the values only into **column D**.
10. Delete the formulas from the **range E5:E33**.


Combine Text

The location names that contain two words were mistakenly split into two columns, so you need to correct this by combining the two columns into one.

- 11.** In **cell E5**, enter the function **=CONCAT(F5:G5)** and then fill the formula down the column.

The cities with two names are missing a space between the two words. To fix this, the **CONCAT** function could be modified, but then a space would be added to each entry even if it isn't necessary. **TEXTJOIN** will work better because a space will only be added when needed.

- 12.** Replace the formula in **cell E5** with the **TEXTJOIN** function:



- 13.** Fill the formula down **column E**; with the **range E5:E33** still selected, copy the formulas and paste the values only into **column F**.
- 14.** Delete the formulas from **column E** (but keep the column) and delete **column G** (remove the entire column).

Use Flash Fill

Now you will fill in the blank columns and create a list of email addresses for the employees.

- 15.** In **cell B5**, enter the name **Amir** and complete the entry.
- 16.** Choose **Data**→**Data Tools**→**Flash Fill**  to fill the first names down column B.
Based on the first example you typed, Amir, Excel uses the text before the space in column A.
- 17.** In **cell C5**, enter the name: **Johnson**
- 18.** Use the **fill handle** in **cell C5** to copy the last name down **column C**, then click the **Auto Fill Options** button and choose **Flash Fill** to replace Johnson with each person's actual last name.

Tip!

Whether you use Flash Fill from the Ribbon or Auto Fill Options, the results will be the same.

The advantage of Flash Fill is that if the result you want is text (not a formula) in each cell, Flash Fill skips the step of inserting a formula before copying and pasting values. The names are now in the appropriate column, so column A can be removed.

- 19.** Delete **column A** entirely.
- 20.** In **cell D4**, enter the heading: **Email**

21. In **cell D5**, enter this email address for Amir: **A.Johnson@airspace.com**
Note that the email address automatically converts to a hyperlink. Although the email address is not fully visible, do not widen the column.
22. In **cell D6**, begin entering Robert's email address by typing **R** and stop when the suggested text appears in column D.
23. Tap **Enter** to accept the suggestions and insert the proper email addresses for all other employees in the column.
24. Save the file.

Creating Conditional Functions Using IF Criteria

Conditional functions allow you to sum, count, and find the average of a range of cells—if the cells meet your desired criteria.

IF CRITERIA FUNCTIONS	
Function	Arguments [Optional]
SUMIF	=SUMIF(range,criteria,[sum range])
AVERAGEIF	=AVERAGEIF(range,criteria,[average range])
COUNTIF	=COUNTIF(range,criteria)
SUMIFS	=SUMIFS(sum range,range1,criteria1,range2,criteria2...)
AVERAGEIFS	=AVERAGEIFS(average range,range1,criteria1,range2,criteria2...)
COUNTIFS	=COUNTIFS(range1,criteria1,range2,criteria2...)

In the preceding table, you may have noticed that a single criterion is entered for the IF functions, and multiple criteria are entered for the IFS functions. Think of IFS as the plural of IF!

IF CRITERIA FUNCTION ARGUMENTS	
Arguments	Description
Range	These are the cells to be compared with the criteria.
Criteria	They can be a comparison value or text, or an expression using a comparison operator such as =, >, <, >=, <=, <> (not equal to).
Sum range/Average range	This is the range to be summed or averaged, which can be different from the range being compared with the criteria. For IF functions, the sum/average range is optional; if omitted, the range from the first argument is used. For IFS functions, the sum/average range comes first and is required.

Using the conditional functions allows you to create formulas to find information such as:

- ▶ How many customers live in Florida?
- ▶ How many employees in the Human Resources division have salaries greater than \$50,000?
- ▶ What are the total sales of product #2152?

For example, if you want to discover the total sales of product #2152 for employees in San Antonio, you would use the SUMIFS function because there are two criteria—the product and the city.

The Sum_range refers to the sales, to be added together from column C.

The formula in cell G3 is shown in the Formula Bar.

The screenshot shows an Excel worksheet with columns A (Product #), B (City), and C (Sales). The data is as follows:

Product #	City	Sales
2151	Detroit	\$5,083
2152	Chicago	\$13,156
2152	San Antonio	\$10,096
2153	Detroit	\$12,684
2153	Chicago	\$5,655
2151	San Antonio	\$7,552

The Formula Bar shows the formula: `=SUMIFS(C3:C12,A3:A12,2152,B3:B12,"San Antonio")`. The Function Arguments dialog box is open, showing the following arguments:

- Sum_range: C3:C12
- Criteria_range1: A3:A12
- Criteria1: 2152
- Criteria_range2: B3:B12
- Criteria2: "San Antonio"

The dialog box also shows a preview of the formula result: 18736.

Criteria1 is the product number 2152; no = sign is needed.

The preview of the formula result is shown here.

Criteria_range1 is the product #s in column A.

Criteria_range2 is the city, in column B.

Criteria2 is "San Antonio", entered in quotations (or Excel adds quotes).

The Function Arguments dialog box makes it easier to enter the arguments because it can be difficult to keep track of the arguments when entering them directly in a cell. The dialog box also shows a preview of the formula result as you add more conditions. The result of the formula above is shown in the worksheet here:

=SUMIFS(C3:C12,A3:A12,2152,B3:B12,"San Antonio")				
D	E	F	G	H
Summary				
	Product	City	Total	
	2152	San Antonio	18736	

DEVELOP YOUR SKILLS: E9-D2

In this exercise, you will use conditional functions to obtain information about the Miami sales team's performance.

1. Save your file as: **E9-D2-SalesAnalysis**

The description for each calculation is located in column K. You will enter the appropriate formula for each in column L.

Use Single Criterion Functions

2. In cell L5, enter this formula: `=COUNTIF(E5:E33,"Miami")`

The formula looks for the criteria Miami and counts each cell that matches this text in the range E5:E33. The formula result shows nine sales employees listed in the Miami location. Next you will find the sum of the Miami sales employees' commissions.

3. In **cell L6**, insert the SUMIF function from the Math & Trig category in the Function Library.
4. In the Function Arguments dialog box, use these arguments:

Range	E5:E33
Criteria	Miami
Sum Range	I5:I33

Once complete, the formula in the Formula Bar shows: =SUMIF(E5:E33,"Miami",I5:I33)

The range and the criteria are the same as the COUNTIF function in step 2, but the SUMIF function also uses the sum range, which are the commissions in the range I5:I33. The result shows \$6,876 in total commissions earned by employees in Miami.

5. Format **cell L6** as **Currency** with no decimals.
Now you will find Miami's average sales.
6. In **cell L7**, enter: =AVERAGEIF(E5:E33,"Miami",H5:H33)

Note!

Remember, you can use the Function Arguments dialog box to enter formulas with several arguments.

This time the range to average is the sales data in column H. The result of the formula is 10643.22222, which you will reformat now.

7. Format **cell L7** as **Currency** with no decimals.

Work with Multiple Criteria Functions

The next thing you want to find is the number of Miami employees who achieved more than \$10,000 in sales.

8. In **cell L8**, insert the COUNTIFS function and open the **Function Arguments** dialog box.
9. Enter **E5:E33** for the first criteria range and **Miami** for the criteria text.
As you insert the arguments, more boxes for additional criteria appear. You can select the ranges from the sheet with the mouse or type them, whatever you prefer.
10. Enter **H5:H33** for the second criteria range and **>10000** for the criteria.
You will leave the third criteria range blank.
11. Click **OK** to enter the function and compare your formula in the Formula Bar to this one:
=COUNTIFS(E5:E33,"Miami",H5:H33,">10000")
The formula searches for Miami in the Location column and numbers higher than 10,000 in the Sales column, and then counts the employees that meet both criteria; the result is 4.
For comparison's sake, you also want to find out how many employees in Toronto had more than \$10,000 in sales.
12. In **cell L8**, enter this formula: =COUNTIFS(E5:E33,"Toronto",H5:H33,">10000")
Use the Function Arguments dialog box, if desired.
The result is only three employees for Toronto. Now you want to find the total commissions paid out to Miami managers who had less than \$10,000 in sales. There will be three conditions this time: Location (Miami), Position (Manager), and Sales (less than \$10,000); the sum range will be Commissions.

13. In **cell L10**, insert the SUMIFS function and use these function arguments:

Function Arguments	
SUMIFS	
Sum_range	I5:I33
Criteria_range1	E5:E33
Criteria1	"Miami"
Criteria_range2	F5:F33
Criteria2	"Manager"
Criteria_range3	H5:H33
Criteria3	"<10000"

The complete formula in the Formula Bar should be:

```
=SUMIFS(I5:I33,E5:E33,"Miami",F5:F33,"Manager",H5:H33,"<10000")
```

This formula finds the sum for commissions if the criteria are met for Location, Position, and Sales.

The result shows that \$2,012 in commissions was paid out to managers in Miami who did not achieve \$10,000 in sales.

14. Format **cell L10** as **Currency** with no decimals.
15. Save the file.

Nested Functions

There are times when you want to perform more than one function, without using two separate cells to do so. In those situations, it is possible to use functions inside of other functions. This is called *nesting* functions, or a *nested* function. Although this can be quite challenging with some functions, it can be fairly simple with others.

For example, when you use the AVERAGE function, you often get a long set of decimal places in the result. While you can adjust the number format, which will change the display of the number, the formula will store those decimal places for future calculations. In some cases, you might want to remove the decimal places altogether from the stored value, and this can be done by using a second function: the ROUND function. Thus, in the same cell, you could nest the AVERAGE function inside of the ROUND function to achieve the desired result.

✕ ✓ fx		=ROUND(AVERAGE(F5:F52),0)
--------	--	---------------------------

This formula finds the average of the range F5:F52 and then rounds that result to zero decimal places.

Another example is, if you need more than one criterion to determine the result for an IF function, and there are more than two possible outcomes, you can nest an IF function in another IF function. For example, you could use the IF function to determine whether an employee achieved a sales goal and then inside that function place another IF function to determine if the employee also achieved a minimum number of sales.

The IFS Function

The IFS function also allows you to work with multiple conditions. With the IFS function, you are able to specify multiple criteria, and the function returns the value for the first one that is true. Depending on the situation, IFS can be used instead of nesting IF functions by rearranging the arguments slightly.

Rather than a value-if-false argument, like the IF function, one of the arguments in the IFS function *must* be true. This means you must carefully write your arguments to include at least one true possibility. Alternatively, the criteria for the last logical test can be entered as TRUE, which will return the corresponding value-if-true no matter what (as long as none of the previous logical tests are true).

The SWITCH Function

New!

The SWITCH function is another logical function that can simplify nested functions in some situations. SWITCH performs an action similar to a lookup and similar to nested IF functions. Essentially the function compares “an expression” to a list and returns the desired result for the matching value. This means it can be used to replace (or “switch”) one thing with another. Like other functions, SWITCH can be combined (nested) with different functions to become even more useful.

For example, the SWITCH function can be used to evaluate the WEEKDAY of a specific date and return the desired results, perhaps the number of employees needed on that day of the week or a short text entry.

	A	B	C	D	E	F	G	H	I
1	Del's Restaurant								
2	Weekly Server Schedule								
3									
4		Date	Feb 18	Feb 19	Feb 20	Feb 21	Feb 22	Feb 23	Feb 24
5		Weekday	M	T	W	Th	F	Sa	Su
6	Name	Staff Needed	3	4	4	6	9	10	CLOSED

In this example, the SWITCH function evaluates the date in row 4 to return a short text entry in row 5 and the number of staff needed in row 6. The functions in cell C5 and C6 are:

- ▶ =SWITCH(WEEKDAY(C4),1,"Su",2,"M",3,"T",4,"W",5,"Th",6,"F",7,"Sa")
- ▶ =SWITCH(WEEKDAY(C4),1,"CLOSED",2,3,3,4,4,4,5,6,6,9,7,10)

Tip!

Remember, the WEEKDAY function examines the date and returns a 1 for Sunday, 2 for Monday, and so on.

Function arguments can be entered in any order, and up to 126 matching values can be entered. At least one value and one result are required, in addition to the expression to evaluate, and a default value can be entered if no matching values are found.

Function Arguments	
SWITCH	
Expression	WEEKDAY(C4) = 2
Value1	1 = 1
Result1	"CLOSED" = "CLOSED"
Default_or_value2	2 = 2
Result2	3 = 3

DEVELOP YOUR SKILLS: E9-D3

In this exercise, you will use a nested function to adjust the result of one formula and the IFS function to calculate an employee bonus in another formula.

1. Save your file as: **E9-D3-SalesAnalysis**
2. Select the Miami Average Sales amount in **cell L7**.
3. Increase the decimal to show three decimal places.

The **AVERAGEIF** function results in a repeating decimal of .222.

4. Follow these steps to edit the formula:



- A In the Formula Bar, click to place the insertion point between the = sign and **AVERAGEIF** and then type: **ROUND(**
- B Click the right side of the Formula Bar and type **,0)** at the end of the formula.
- C Click **Enter** on the Formula Bar to complete the changes.

The result displayed now shows \$10,643.000, which is the average sales rounded to the nearest dollar.

5. In **cell L7**, decrease the decimal to remove the decimals from the display.

Using the IFS Function

6. Select **cell J4** and insert a new worksheet column.
7. In **cell J4**, apply a thick outside border and enter the heading: **Bonus**
Because the commissions earned on sales can vary, you offer a 1% bonus to employees who met their sales target and a 2% bonus to employees whose commissions were also less than \$1,000.
8. In **cell J5**, insert the **IFS** function from the Logical category of the Function Library.

Note!

The order of the logical tests is very important, since the function returns the value for the first test that is true!

9. Follow these steps to create the IFS function:

Function Arguments			
IFS			
Logical_test1	H5<G5	↑	= FALSE
Value_if_true1	0	↑	= 0
Logical_test2	I5>=1000	↑	= FALSE
Value_if_true2	1%*H5	↑	= 50.83

Function Arguments			
IFS			
Logical_test3	I5<1000	↑	= TRUE
Value_if_true3	2%*H5	↑	= 101.66

- A** In the first logical test, enter **H5<G5** to determine which employees did not meet their target.
- B** The Value_If_True for employees who did not meet their target is no bonus, so enter: **0**
- C** If the first test is false, the employee did meet their target, so enter **I5>=1000** in the second logical test to determine if their commissions were greater than or equal to \$1,000.
- D** If the second test is true, the employee did meet their goal but had commissions greater than \$1,000, so call for a 1% bonus by typing: **1%*H5**
- E** In the third logical test type: **I5<1000**
- F** Type **2%*H5** as the Value_If_True formula so employees who did meet their goal and had commissions less than \$1,000 will receive a 2% bonus.
- 10.** Click **OK** to enter the formula, modify the number format to show two decimals, and then fill the formula down the column for all other employees.
The result for the first employee is a bonus of \$101.66.
- 11.** Apply all borders to the **range J5:J33** to match the rest of the data and then reapply a thick outside border to **cell J4**.

Using the SWITCH Function

The company decided to modify the job titles, so you will use SWITCH to replace the existing names with new ones.

- 12.** Select **cell G5** and insert a new column.

13. In **cell G5**, insert the **SWITCH** function with these arguments:


Function Arguments			
SWITCH			
Expression	F5	↑	= "Agent"
Value1	"Agent"	↑	= "Agent"
Result1	"Sales Rep"	↑	= "Sales Rep"
Default_or_value2	"Manager"	↑	= "Manager"
Result2	"Sales Director"	↑	= "Sales Director"

If cell F5 contains Agent it will be replaced with Sales Rep, and if it contains Manager it will be replaced with Sales Director.

14. Fill the formula down for all employees, then copy the new positions from **column G** and paste the values only into **column F**; delete **column G**.
15. Save your work.

Troubleshooting Formulas

As you might have noticed, working with formulas can sometimes be complicated. Excel's **auditing** tools can help you make sense of your worksheet when it contains a complex set of formulas. The auditing tools can help you identify which cells were used to create a formula or where a particular cell is being used in other formulas, as well as locate and correct errors in formulas.



Trace Precedents and Dependents

The **Trace Precedents** command displays arrows pointing to the current active cell from any cells that were used to produce the result. Trace Precedents works backward from the selected cell to show which cells affect the current result. Because the precedent cells could also use input from other cells to produce their results, there can be several layers of precedents. Repeating the Trace

Precedents command will display the next level of precedents until a warning sound indicates there are no more levels.

Name	Goal	Sales	# Sales
Bert	\$ 1,000	\$ 900	18
Ernie	1,200	1,300	12
Jen	800	950	21
Sarah	1,000	1,200	17
Total	\$ 4,000	\$ 4,350	68
Sales Above Goal		\$ 350	
Average Sale		\$ 63.97	

With the current cell displaying the Sales Above Goal amount, Trace Precedents shows that the total Goal and Sales cells are used to calculate the above-goal amount of \$350.

Name	Goal	Sales	# Sales
Bert	\$ 1,000	\$ 900	18
Ernie	1,200	1,300	12
Jen	800	950	21
Sarah	1,000	1,200	17
Total	\$ 4,000	\$ 4,350	68
Sales Above Goal		\$ 350	
Average Sale		\$ 63.97	

Adding another level to Trace Precedents shows that the totals use the information in the Goal and Sales columns.

Although you can see which cells are used in a formula by looking at the formula in the Formula Bar, tracing precedents is much quicker and gives you a better way to visualize the flow of data through the worksheet.

The **Trace Dependents** command shows the opposite of the precedents; it shows you any cells that use the current cell in a formula. Like precedents, there can be layers of dependent cells, which are displayed by repeating the Trace Dependents command. Changing the value in the current cell will therefore have an effect on all of the dependent cells.

Name	Goal	Sales	# Sales
Bert	\$ 1,000	\$ 900	18
Ernie	1,200	1,300	12
Jen	800	950	21
Sarah	1,000	1,200	17
Total	\$ 4,000	\$ 4,350	68
Sales Above Goal		\$ 350	
Average Sale		\$ 63.97	

The current cell shows Bert's number of sales, and by using Trace Dependents you see that Bert's sales amount is used to calculate total sales.

Name	Goal	Sales	# Sales
Bert	\$ 1,000	\$ 900	18
Ernie	1,200	1,300	12
Jen	800	950	21
Sarah	1,000	1,200	17
Total	\$ 4,000	\$ 4,350	68
Sales Above Goal		\$ 350	
Average Sale		\$ 63.97	

By adding another level, you see that total sales is then used to calculate the average sale.

Another way to think of it is to think of tracing precedents as looking backward, to see where the information comes from, and tracing dependents as looking forward, to see where the information is being used. When you no longer need the arrows, you can simply use the Remove Arrows command to remove them.



View the video “Tracing Your Formulas.”

Checking for Errors

The Error Checking tool can help you spot and correct errors in formulas. This can be particularly useful if you are reviewing someone else’s work and aren’t sure where the errors are located. If it is your own work, you would usually fix errors as you go along.

Errors in cells are flagged with a green triangle, and sometimes with an error message in the cell instead of the formula result.

A green triangle indicates a cell that contains an error.

Selecting a cell with an error displays the warning sign.

Clicking the warning sign displays options for dealing with the error.

Note!

Even if the formula displays a result, it might still contain an error, usually if the formula omits adjacent data. If this is done intentionally, you can click the warning sign and select Ignore Error to dismiss it.

COMMON EXCEL FORMULA ERRORS	
Error	Description
#DIV/0!	Dividing by zero is not possible, so this error displays if a formula attempts to divide by a cell that contains zero or by an empty cell.
#REF!	You will see this error if a formula contains an invalid cell reference; for example, if a formula refers to cell A1 and row 1 or column A is deleted.
#VALUE!	This error usually occurs because the formula is attempting to perform a mathematical operation using a cell that contains text.
#NAME?	This error occurs when a formula contains an incorrect function name or an undefined name for a cell or range.
Formula Omits Adjacent Cells	This error does not display in the result cell. It appears as a suggested error when a formula refers to a column or row of data but does not include all adjacent numerical values.

Evaluate a Formula

Not all mistakes result in Excel displaying an error; sometimes a valid formula is simply showing an incorrect result because it was not created with the correct cell references, functions, or operations. When reviewing a complex formula, it can be useful to break the formula down into steps and watch how Excel solves it. This can help you discover the source of an error or explain why the result does not look the way you expected. Evaluating a formula can also help you see how a formula with multiple operations is solved (by following the correct order of operations one step at a time).

The Watch Window

The Watch Window allows you to keep track of a particular formula, even when working on a different sheet or workbook. This is useful when you have multiple sheets with formulas that use data across many sheets, and you want to observe the effects of changes made on one sheet to the results of formulas on other sheets.

☰ Formulas → Formula Auditing

DEVELOP YOUR SKILLS: E9-D4

In this exercise, you will use the Formula Auditing tools to analyze formulas and correct formula errors on the Aug Sales sheet.

1. Save your file as: **E9-D4-SalesAnalysis**
2. Go to the **Aug Sales** sheet and select **cell M10**.
3. Choose **Formulas** → **Formula Auditing** → **Trace Precedents** .

You will now see lines drawn from the Location, Position, Sales, and Commissions columns, all pointing to cell M10. Because the formula refers to the range within the column, there is also a blue box around the entire range.

Location	Position	Target	Sales	Commissions	Bonus		Description	Amount
Miami	Sales Rep	\$ 5,000	\$ 5,286	\$ 589	\$ 105.72		Miami Employees	9
Miami	Sales Rep	10,000	13,419	1,251	\$ 134.19		Miami Employee Commissions	\$8,390
New York	Sales Rep	10,000	10,298	883	\$ 205.96		Miami Average Sales	\$10,890
New York	Sales Director	12,000	14,713	535	\$ 294.26		Miami Employees - Sales >\$10,000	4
Toronto	Sales Rep	5,000	5,485	437	\$ 109.70		Toronto Employees - Sales >\$10,000	4
Vancouver	Sales Director	10,000	5,815	614	\$ -		Miami Mgr Comm's - Sales <\$10,000	\$0

Tracing precedents also shows you that the position name has changed and needs to be updated in the formula.

4. In the Formula Bar, modify the formula for **cell M10** by replacing *Manager* with *Sales Director*:
=SUMIFS(I5:I33,E5:E33,"Miami",F5:F33,"Sales Director",H5:H33,"<10000")

The result in cell M10 displays \$2,427. Modifying the formula also removes the arrows.

5. With **cell M10** still selected, choose **Formulas** → **Formula Auditing** → **Trace Precedents**  two times.

The first click shows the precedents again. On the second click, no arrows are added because the Location, Position, Sales, and Commissions columns all contain values—not formulas. If your volume is turned on, you may hear the warning sound indicating the command can't be completed.

6. With **cell M10** still selected, choose **Formulas**→**Formula Auditing**→**Trace Dependents** .

You will see a warning box indicating there are no formulas that refer to the active cell.

7. Click **OK** to close the warning box.

8. Choose **Formulas**→**Formula Auditing**→**Remove Arrows** .

9. Select **cell M18** and choose **Formulas**→**Formula Auditing**→**Trace Precedents** .

The Trace Precedents arrows indicate that cells M15, M16, and M17 are used in the formula to calculate the total paid out. Because the Sales amount is not paid to employees, this should not be included, and the formula will have to be corrected.

10. Edit the formula in **cell M18** to be: **=M16+M17**

11. With **cell M18** still selected, click the **Trace Precedents**  button twice.

The next level of precedents appears, showing that the formulas to calculate total commissions and bonuses use references to the data in the Sales and Commissions columns. This should be a warning that the formulas might be incorrect because Total Bonus does not refer to the Bonus column of data, and the ranges also do not include the entire column as they should.

12. Click the **Trace Dependents**  button.

The dependent arrow points to cell M19.


Check for Errors

Now you will find the errors on the sheet and correct them.

13. Remove the arrows from the worksheet.

14. Select **cell A1** and choose **Formulas**→**Formula Auditing**→**Error Checking** .

The Error Checking dialog box opens and displays the first error in cell M15. The formula is =SUM(H5:H17), and the error is that the formula omits the adjacent cells.



Because Excel anticipates that formulas will normally refer to an entire column or row and not just part, the error asks whether you want to update the formula to include cells. Of course, if the formula is meant to use only part of the column, you can ignore the error; in this case, you will correct it.

15. Click **Update Formula to Include Cells**.

The error checker moves to the next error in cell M16. The error in cell M16 is the same error; the formula does not include all cells in the column.

16. Click **Update Formula to Include Cells** again.

The error checker moves to the next error in cell M17. The error in cell M17 also omits adjacent cells; however, the formula also refers to the wrong column, so you can fix this by directly editing the formula.

17. Click **Edit in Formula Bar**.

18. In the Formula Bar, replace the existing range reference with **J5:J33** so the formula reads **=SUM(J5:J33)** and then click **Resume** in the Error Checking dialog box.

This time the error in cell M19 is an error in value, meaning a value used in the formula is of the wrong data type, usually text instead of a number. The formula is **=M18/M14**, but cell M14 contains text. You can edit this formula in the Formula Bar so the formula is dividing Total Paid Out/Total Sales.

19. Click **Edit in Formula Bar**, then correct the formula to **=M18/M15** and click **Resume**.

The rest of the worksheet is now error free.

20. Click **OK** in the dialog box and apply the **Percent Style** number format to **cell M19**.

Evaluate a Formula

21. With **cell M19** still selected, choose **Formulas**→**Formula Auditing**→**Evaluate Formula** .

Evaluate Formula	
Reference:	Evaluation:
'Aug Sales'!\$M\$19	= M18/M15

The Evaluate Formula dialog box shows the reference to the cell being evaluated, cell M19 on the Aug Sales sheet, and the formula being evaluated, **=M18/M15**. The underline indicates the next part of the formula to evaluate.

22. Click **Evaluate**.

Evaluate Formula	
Reference:	Evaluation:
'Aug Sales'!\$M\$19	= 26800.71/M15

Cell M18 is evaluated, and the value of cell M18 is displayed: 26800.71. Note that the cell's true value is shown, not the value displayed on the worksheet, which is affected by the number format (in this case no decimals). The second cell reference, M15, is underlined, indicating it is the next part of the formula that will be evaluated.

23. Click **Evaluate**.

Evaluate Formula	
Reference:	Evaluation:
'Aug Sales'!\$M\$19	= 26800.71/319535

Now the value of cell M15 is displayed: 319535.

24. Click **Evaluate** once more.

The result of the formula 26800.71/319535 is now shown, which is 8%.

25. Click **Close** in the Evaluate Formula dialog box.

26. Save the file.

3-D Cell References

Excel formulas can refer to data on other worksheets or in other workbooks; however, sometimes a formula is required to refer to multiple sheets at the same time. For example, if cell A5 contains sales information for Product #1, and there is a different sales worksheet for each month, you might want to summarize the sales data by finding the total in cell A5 across multiple sheets. This can be done by adding each cell individually or by using a **3-D reference** in your formula.

Compare these two formulas.

- ▶ =SUM(January!A5+February!A5+March!A5+April!A5+May!A5)
- ▶ =SUM(January:May!A5)

Similar to using a range instead of referring to each cell individually, a 3-D reference is quicker because you can refer to the range of sheets January to May and all sheets between, and then use the same cell (or range) from all sheets.

If you think of columns and rows as arranged left to right and up and down on a 2-D page, then stacking multiple worksheets on top of each other would be a third dimension, hence the 3-D reference. This is also useful because if a new sheet is inserted into the worksheet range, the data on the new sheet will automatically be included.

DEVELOP YOUR SKILLS: E9-D5

In this exercise, you will use 3-D cell references to create formulas that sum values from three worksheets.

1. Save your file as: **E9-D5-SalesSummary**
2. Go to the **Q3 Sales Summary** worksheet.
For each employee you will summarize the total target, sales, and commission amounts for July, August, and September.
3. In **cell G5**, begin entering your formula by typing: =SUM(
4. Follow these steps to finish the formula using a 3-D reference:



- A** While editing the formula, click the **Jul Sales** worksheet tab.
- B** Click **cell G5** and see that the Formula Bar displays =SUM('Jul Sales'!G5).
- C** Hold down **[Shift]**+click the **Sept Sales** sheet tab to select the range of sheets Jul Sales:Sept Sales, which includes the Aug Sales sheet.
5. Complete the entry by clicking on the Formula Bar, which automatically returns you to the Q3 Sales Summary worksheet.

Warning!

Be careful when completing the entry. Do NOT click the Q3 Sales Summary tab or any other cells, as this will change the formula!

The completed formula is =SUM('Jul Sales:Sept Sales'!G5), which finds the sum of cell G5 on all three sheets. The result of the formula is \$15,000.

6. Select **cell H5** and complete steps 3–5 again to create a formula that adds cell H5 on the Jul Sales, Aug Sales, and Sept Sales sheets.
Your completed formula should be =SUM('Jul Sales:Sept Sales'!H5).
 7. In **cell I5**, create a formula that adds cell I5 on the Jul Sales, Aug Sales, and Sept Sales sheets.
Your completed formula should be =SUM('Jul Sales:Sept Sales'!I5). The sum in cell H5 is \$16,113, and the sum in cell I5 is \$1,322.
 8. Select the **range G5:I5** and use the **fill handle** to copy the formulas down all three columns.
 9. Save your work and close the file.
-

Self-Assessment



Check your knowledge of this chapter's key concepts and skills using the Self-Assessment in your ebook or online (eLab course or Student Resource Center).

Reinforce Your Skills

REINFORCE YOUR SKILLS: E9-R1

Work with Text Functions and Conditional Functions

In this exercise, you will take the information that was recorded from the Kids for Change Charity Race and fix the participants' names and bib numbers. You will then analyze the race results.

1. Open **E9-R1-FunRun** from your **Excel Chapter 9** folder and save it as: **E9-R1-RaceResults**
2. Select **column B** and insert a new column.
3. In **cell B5**, enter the formula: **=PROPER(A5)**
The formula takes the text in cell A5 and capitalizes the first letters of the first and last names.
4. Copy the formula down the column using the **fill handle**.
5. With the **range B5:B22** selected, copy and paste the names into **column A** and be sure to paste the values only.
6. Delete **column B**.
Now you will alter the bib numbers by removing the text to display the digits only.
7. Select **column E** and insert a new column.
8. In **cell E5**, enter: **410**
9. Use the **fill handle** to copy **cell E5** down the column and then use the Auto Fill Options to choose **Flash Fill**.
Flash Fill enters the three numbers from the Bib # for all runners.
10. In **cell E4**, type **Bib #** and then delete **column D**.

Conditional Functions

11. In **cell J5**, enter: **Runners 40 & Over**
12. In **cell K5**, enter the formula: **=COUNTIF(B5:B22,">=40")**
Use the Function Arguments dialog box, if desired, to enter the arguments one at a time. The result shows there were four runners aged 40 or older.
13. In **cell J6**, enter: **~Average Time**
14. In **cell K6**, enter the AVERAGEIF function with these arguments:

Range	B5:B22
Criteria	>=40
Average_Range	H5:H22

The completed formula should look like this: =AVERAGEIF(B5:B22,">=40",H5:H22)

The formula finds the average time in the range H5:H22 for runners who are 40 or older in the range B5:B22. Now the result needs to be formatted as time.

15. With cell K6 still selected, modify the number format to show hours and minutes only.
Now you can see that the average time for the four runners who are 40 or older is one hour and forty minutes.

16. In **cell J7**, enter: **~Total Raised**
17. In **cell K7**, enter the formula: **=SUMIF(B5:B22,">=40",E5:E22)**
18. With **cell K7** still selected, format the result as Currency with no decimals.
The total raised by the 40 and over group was \$770.
19. In **cell J8**, enter: **Runners Under 25**
20. Copy and paste the contents of the **range J6:J7** into the **range J9:J10**.
21. Enter appropriate formulas into the **range K8:K10** to count the runners under 25, calculate their average time, and determine the total amount they raised.
Use the previous steps to guide you, if necessary, as the arguments will be similar, except the criteria will be <25 this time. Hint: Take a look at the functions here for more guidance:

	J	K
8	Runners Under 25	=COUNTIF(B5:B22,"<25")
9	~Average Time	=AVERAGEIF(B5:B22,"<25",H5:H22)
10	~Total Raised	=SUMIF(B5:B22,"<25",E5:E22)

22. Use the **Format Painter** to copy the format from the **range K6:K7** to the **range K9:K10**.

Race Results	
Description	Result
Runners 40 & Over	4
~Average Time	1:40
~Total Raised	\$770
Runners Under 25	5
~Average Time	3:06
~Total Raised	\$1,012

23. Save your work and close the file.

REINFORCE YOUR SKILLS: E9-R2

Use Nested Functions and 3-D Cell References

In this exercise, you will use functions to summarize data from five cities where Kids for Change held their charity races.

- Open **E9-R2-FunRun** from your **Excel Chapter 9** folder and save it as: **E9-R2-RaceSummary**
- Select **cell C6** on the **Summary** sheet and create a formula to add all of the participants from **cell C5** on the five city sheets using a 3-D reference.
Be sure to select the cell and worksheets with the mouse and do not simply type the formula. The finished formula will be =SUM('New York:Washington!C5).
- Using the **fill handle**, copy the formula into the **range C7:C9**, which will find the sum from all five cities for volunteers, expenses, and funds raised.
The formula for net contributions is already inserted in cell C10, so the formula is updated with the data in cells C8 and C9.
- Apply the Currency number format and remove the decimals from the **range C8:C9**.
- Select **cell C10**, which contains the net contributions raised from all five cities.

6. Choose **Formulas**→**Formula Auditing**→**Trace Precedents** .

The arrows indicate that cell C10 uses the information in cells C8 and C9.

7. Click **Trace Precedents**  a second time.

This shows the next level of precedents, indicating that cells C8 and C9 use data from other worksheets.

8. Click the **Remove Arrows**  button.

In cell C11 you will create a formula to determine whether all cities achieved the minimum goal of raising \$35,000, using a nested function with a 3-D cell reference.

9. Select **cell C11** and use the **Insert Function** button to insert the IF function.

The logical test for the IF function arguments will use the MIN function.

10. In the Logical_Test box, type **MIN(** and then select the **New York** sheet. Select **cell C8**, and then hold down **[Shift]** and select the **Washington** sheet.

11. Type **)** to close the MIN function parentheses and then complete the logical test by typing: **>=35000**

The logical test will find the lowest number for the funds raised across all five cities and then test whether that number is greater than or equal to \$35,000.

12. Enter **Yes** in the Value_If_True field and **No** in the the Value_If_False field.

Function Arguments			
IF			
Logical_test	MIN('New York:Washington'!C8		= TRUE
Value_if_true	"Yes"		= "Yes"
Value_if_false	"No"		= "No"
			= "Yes"

The Logical_Test field does not display the entire argument; only the first part is visible due to the limited size. The complete formula is =IF(MIN('New York:Washington'!C8)>=35000,"Yes","No").

13. Click **OK**.

The result of the formula is Yes, which means all five cities either met or surpassed the goal.

14. Right-align **cell C11**.

15. In **cell C12**, enter the formula: **=C6/C5**

The error appears because cell C5 has not been calculated yet; it is empty, so dividing by an empty cell returns the #DIV/0! error.

16. Select **cell C5**; begin to enter the formula **=COUNT(** and then use a 3-D reference to count **cell C5** on all the city sheets to count the number of cities involved.

You can see there are five cities; however, if another city is added and a new sheet is inserted into the range, the formula results in cell C5 and cell C12 will automatically update, rather than require manual editing.

The error is now removed from cell C12, and the result shows each city had an average of 189.6 participants.

17. Save your work and close the file.

REINFORCE YOUR SKILLS: E9-R3

Analyze Volunteer Data

In this exercise, you will look at the number of hours Kids for Change volunteers have donated over the past three years and use formulas to analyze and summarize the data.

1. Open **E9-R3-Vhours** from your **Excel Chapter 9** folder and save it as:
E9-R3-VhoursSummary
2. In **cell B5**, type **Dave Lozano** to put the volunteer's surname after his given name and tap **Enter**.
3. Type **S** (the first letter of Sharon's first name) in **cell B6**, and tap **Enter** to accept the Flash Fill suggestions, entering the full names for all volunteers in the same order.
4. Cut and paste the contents of the **range B5:B11** into the **range A5:A11**.
5. In **cell B5**, enter a formula that will add all hours for Dave from 2018 to 2020.
Hint: Use the SUM function and use the mouse to insert a 3-D reference to the range B5:M5 on all three worksheets.
Across three years, Dave has volunteered 596 total hours.
6. In **cell C5**, enter a formula that will count all months in 2020 in which Dave volunteered more than 25 hours.
Hint: Use the COUNTIF function, insert a reference to the range B5:M5 on the 2020 sheet, and use ">25" as the criteria.
The result shows there were five months in 2020 in which Dave volunteered more than 25 hours.
7. In **cell D5**, enter a formula similar to the one created in **cell C5**, but modify the criteria to count the number of months in which Dave had zero hours.
The result shows there were two months in 2020 in which Dave did no volunteer work.
Now you want to find which volunteers averaged more than 15 hours per month in 2020, so you will use the AVERAGE function nested inside an IF function.
8. In **cell E5**, begin by inserting an IF function and open the Function Arguments dialog box.
9. In the Logical_Test field, use: **AVERAGE('2020'!B5:M5)>15**
Hint: Use the mouse to select the range for the average function.
10. Enter **Yes** in the Value_If_True field and **No** in the Value_If_False field and then click **OK**.
11. Right-align **cell E5**.
12. Select the **range B5:E5** and copy the formulas in **row 5** down to **row 11** for the other volunteers.
The results show that only one volunteer did not average more than 15 hours per month in 2020.
13. Save your work and close the file.

Apply Your Skills

APPLY YOUR SKILLS: E9-A1

Create a Data Summary

In this exercise, you will summarize expense data for two of Universal Corporate Event's sales employees, so that other employees can be inserted later on and their information will instantly be added to the total.

1. Open **E9-A1-Expenses** from your **Excel Chapter 9** folder and save it as: **E9-A1-ExpenseReport**
First you need to capitalize the expense names on all sheets.
2. In **cell A10**, enter a formula to convert the text in **cell A5** to PROPER text format.
3. Copy the formula down to **cell A13** to convert all four expenses.
4. Copy and paste the **range A10:A13** into the **range A5:A8** on all three sheets.
Hint: Paste the values only; copy once and then paste three times.
5. Delete the contents of the **range A10:A13** on the **Summary** sheet.
Now you will combine the account type code with the expense account number.
6. Enter a formula in **cell C10** that uses the CONCAT function to combine **cell C5** and **cell B5**, in that order.
The expense type code now precedes the expense account number.
7. Copy the formula in **cell C10** down to **cell C13**.
8. Copy and paste (values only) from the **range C10:C13** into the **range B5:B8**.
9. Delete **column C**.
10. Copy the **range B5:B8** from the Summary sheet into the same range on the two expense sheets for **David** and **Maria**, and then delete **column C** on both sheets.
11. In **cell C5** on the **Summary** sheet, create a formula with the SUM function that uses a 3-D reference to add both employees' January travel expenses.
12. Copy the formula down the column for all expenses and then copy the same four formulas across the rows for all months plus the total.
13. Format the total in **column O** of all three sheets with the Accounting number format, bold formatting, and no decimals. (Tip: If desired, select all three sheets and then edit them simultaneously.)
14. AutoFit **columns A–O** on all three sheets.
15. Save your work.

APPLY YOUR SKILLS: E9-A2

Evaluate Expenses

In this exercise, you will analyze the meal expenses for employees to see how much you would save if you were to cap monthly meal expenses that could be claimed.

1. Save your file as: **E9-A2-ExpenseReport**
2. In **cell A11** on the **Summary** sheet, enter: **Meals Exp >500**

3. Enter **Count** in **cell A12** and **Sum** in **cell A13**.
4. In **cell B12**, enter a formula using the COUNTIF function that will find the number of months in which employees' meals exceeded \$500.
5. In **cell B13**, enter a formula using the SUMIF function to find the sum of the meals for the months in which employees' meals exceeded \$500.
6. Enter **\$ Above \$500** into **cell A14**, apply a thick outside border, center-align the contents, and then copy and apply the same formatting to **cell A11**.
7. In **cell B14**, enter the formula: **=B13-(B12*500)**
This formula takes the sum from the months over \$500 and subtracts the number of months multiplied by \$500 to find the difference that would be saved if there were a \$500 cap on meal expenses.
8. Apply bold formatting and the Accounting number format to **cell B14**.
9. With **cell B14** still selected, use the **Trace Precedents**  command as necessary until there are no more precedents to show.
10. Save your work.
Notice the trace precedent arrows are removed.
11. In **cell D9**, enter the formula **=IF(SUM(D5:D8)>SUM(C5:C8),"+","-")** and then copy the formula across the row into the **range E9:N9**.
This formula compares the sum of each month's expenses with the sum of the previous month's expenses. If they are greater, the result will be a plus sign and if they are less than or equal, the formula will return the minus sign; this will show upward or downward trends for overall expenses.
12. With the formula results in the **range D9:N9** still selected, apply bold formatting, center-align the cells, and increase the font size to 12 points.
13. Increase the row height of **row 11** to **32**.
14. Save your work and close the file.

APPLY YOUR SKILLS: E9-A3

Analyze Sales

In this exercise, you will modify the client codes for the Universal Corporate Events client list and then compare sales results for training events sold by each employee.

1. Open **E9-A3-Q3** from your **Excel Chapter 9** folder and save it as: **E9-A3-Q3Analysis**
To begin, you have been asked to add the first three letters of each client's name to the client code in all capital letters. To do this, you can use the LEFT function, nested inside the UPPER function.
2. In **cell G5**, create a formula using the LEFT function to extract the first three letters of the client name in **cell A5**.
3. Modify the formula to put all three letters in caps by placing the LEFT function inside the UPPER function.
The formula should be =UPPER(LEFT(A5,3)).
4. In **cell H5**, enter a formula using the CONCAT function to combine **cell G5** with **cell B5** (in that order).
5. Copy **cells G5** and **H5** down the columns to **row 16** using the **fill handle**.
6. Copy and paste (values only) from the **range H5:H16** into the **range B5:B16**.

7. Delete the contents in **columns G** and **H**.
8. In **cell B19**, enter a formula that counts the number of training events sold by David.
Hint: The ranges C5:C16 and E5:E16 will be your criteria ranges.
9. In **cell C19**, enter a formula that adds the fees for training events sold by David.
Hint: The criteria ranges will be the same as those used in the COUNTIFS function in step 8 and the sum range will be F5:F16.
10. In **cells B20** and **C20**, enter formulas to count and sum again, but this time use criteria to find data for the training events sold by Maria.
11. Apply the Currency number format to the **range C19:C20** and remove the decimals.
12. Save your work and close Excel.

EVALUATION ONLY

Project Grader

If your class is using eLab (labyrinthelab.com), you may upload your completed Project Grader assignments for automatic grading. You may complete these projects even if your class doesn't use eLab, though you will not be able to upload your work.

PROJECT GRADER: E9-P1

Updating Distributor Data Using Text Functions, Conditional Functions, and Formula Auditing

Taylor Games is updating its distributor worksheets to display sales by distributor. In this exercise, you will create formulas that will capitalize the last names of distributors, create a distributor ID using first and last name data, and calculate annual sales using a 3-D reference. You will then compare these sales with the annual sales goal using IF functions.

- Download and open your Project Grader starting file.
 - Using eLab: Download **E9_P1_eStart** from the Assignments page. You must start with this file or your work cannot be automatically graded.
 - Not using eLab: Open **E9_P1_Start** from your **Excel Chapter 9** folder.
- In **cell J7** of the **Distributors** worksheet, use the PROPER function to reference **cell B7**, creating a capitalized copy of the last name (*Gilson*).
- Copy the formula in **cell J7** to the **range J8:J36**.
- Copy the **range J7:J36** and paste only the values in the **range B7:B36** to replace the last names in column B with properly formatted replacements.
- Hide **column J**.
- In **cell D7**, combine the first name and last name from **cells A7** and **B7** using the CONCATENATE function with the following nested arguments:
 - Argument1: Use the UPPER function to convert the text in **cell B7** to uppercase.
 - Argument2: Use the LEFT function to extract the first letter of the first name from **cell A7**.

Your nested functions should produce the following result in cell D7:

First Name	Last Name	Distributor #	Distributor ID
Carey	Gilson	453678	GILSONC

- Copy the nested functions to the **range D8:D36**.
- Use these guidelines to create a function with a 3-D cell reference:
 - Use the SUM function in **cell E7**.
 - Use a 3-D reference to add up the values in **cell B7** in all four **Distributor Sales** worksheets (**Distributor Sales Q1:Distributor Sales Q4**).
- Change the number format of **cell E7** to Accounting and apply bold formatting.
- Copy the formula in **cell E7** to the **range E8:E36**.
- In **cell I3**, use the COUNTIF function to determine the number of distributors in the **range E7:E36** with annual sales that are greater than or equal to \$1,000.

12. In **cell I4**, use the COUNTIF function to determine the number of distributors with annual sales that are less than \$1,000.
13. Save your workbook.
 - *Using eLab:* Save it to your **Excel Chapter 9** folder as **E9_P1_eSubmission** and attach the file to your eLab assignment for grading.
 - *Not using eLab:* Save it to your **Excel Chapter 9** folder as: **E9_P1_Submission**

PROJECT GRADER: E9-P2

Calculating Membership Car Value

Classic Cars Club would like to evaluate the types of cars and corresponding values for their members. In this exercise, you will create a club profile using single and multiple criteria functions to display the amount and values of classic and antique cars. You will also generate email addresses for each member based on the member's name and the club's email domain.

1. Download and open your Project Grader starting file.
 - *Using eLab:* Download **E9_P2_eStart** from the Assignments page. You *must* start with this file or your work cannot be automatically graded.
 - *Not using eLab:* Open **E9_P2_Start** from your **Excel Chapter 9** folder.

Note: When creating conditional functions in steps 2–11, you can use either ranges in your arguments or reference the Car Values table. For example, using the range F8:F37 will produce the same outcome as referencing the table CarValues[Status]. Feel free to use either approach.

2. Enter a formula in **cell I9** of the **Car Values** worksheet that will count the number of cars with a status of *classic*.
3. Enter a formula in **cell J9** that will count the number of cars with a status of *antique*.
4. Enter a formula in **cell I11** that will count the number of cars with a status of *classic* and a car value greater than \$100,000.
5. Enter a formula in **cell J11** that will count the number of cars with a status of *antique* and a car value greater than \$100,000.
6. Enter a formula in **cell I13** that will count the number of cars with a status of *classic* and a car value less than \$50,000.
7. Enter a formula in **cell J13** that will count the number of cars with a status of *antique* and a car value less than \$50,000.
8. Enter a formula in **cell I15** that will average the car values for all cars with a status of *classic*.
9. Enter a formula in **cell J15** that will average the car values for all cars with a status of *antique*.
10. Enter a formula in **cell I17** that will add up the car values for all cars with a status of *classic*.
11. Enter a formula in **cell J17** that will add up the car values for all cars with a status of *antique*.

- 12.** In **cell D7** of the **Membership List** worksheet, use these guidelines and a CONCATENATE function to turn names into email addresses:
- Argument1: Use a nested LEFT function to extract the first letter of the first name from **cell A7**.
 - Argument2: Reference **cell B7** to extract the entire contents of the cell.
 - Argument3: Enter the literal text string: "**@ClassicClub.Org**"

Your formula should produce the following result in cell D7:

First Name	Last Name	Membership #	Email
Betsy	Lindburgh	CC-32467	BLindburgh@ClassicClub.Org
Loretta	Lemon	CC-56478	

- 13.** Copy the formula in **cell D7** and paste it to the **range D8:D36**.
- 14.** Save your workbook.
- Using eLab: Save it to your **Excel Chapter 9** folder as **E9_P2_eSubmission** and attach the file to your eLab assignment for grading.
 - *Not using eLab:* Save it to your **Excel Chapter 9** folder as: **E9_P2_Submission**

Extend Your Skills

These exercises challenge you to think critically and apply your new skills in a real-world setting. You will be evaluated on your ability to follow directions, completeness, creativity, and the use of proper grammar and mechanics. Save files to your chapter folder. Submit assignments as directed.

E9-E1 That's the Way I See It

After tracking your personal budget for six months, you want to analyze your spending. Open **E9-E1-Budget** and save it as: **E9-E1-BudgetAnalysis**

Fill the Budget Total and Actual Total formulas down the respective columns and then use the Error Checking feature to remove any errors from the worksheet (except those found in column J). Then, in the range M5:M11, use the COUNTIF function to count the number of months from January to June that exceeded the budget amount in column B. (Hint: Do not use a cell reference for the criteria; type in the budget amount.) Again, use Error Checking as necessary to remove any errors.

E9-E2 Be Your Own Boss

It's year-end, and you're analyzing the annual income at Blue Jean Landscaping. You've already calculated the revenue by month, and now you're ready to find the annual totals. Open **E9-E2-Revenue** and save it as: **E9-E2-RevenueSummary**

There are four quarterly worksheets. On the Summary sheet, use a SUM function in column C that adds each of the individual months on the four quarter sheets. After you have the annual totals calculated for both the Private and Corporate divisions, use a formula in cell D15 to tell you in one step which division had higher revenue. Use a nested IF function; for the logical test, compare the sum of private revenue with the sum of corporate revenue (using the SUM function twice). Return either *Private* or *Corporate* based on the result.

E9-E3 Demonstrate Proficiency

Stormy BBQ has expanded BBQ sauce sales to grocery stores in several cities, and you want to find out where the different flavors are selling the best. Open **E9-E3-Sales** and save it as: **E9-E3-SalesComparison**

Use the SUMIFS function to fill in the table in the range I5:K7, using two criteria each time: Sauce Type and Location. In the range J10:J12, use COUNTIFS to find the number of times each city placed an order for \$3,000 or more.